

# NAG Fortran Library Routine Document

## F01RJF

**Note:** before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

### 1 Purpose

F01RJF finds the  $RQ$  factorization of the complex  $m$  by  $n$  ( $m \leq n$ ), matrix  $A$ , so that  $A$  is reduced to upper triangular form by means of unitary transformations from the right.

### 2 Specification

```
SUBROUTINE F01RJF(M, N, A, LDA, THETA, IFAIL)
INTEGER          M, N, LDA, IFAIL
complex        A(LDA,*), THETA(*)
```

### 3 Description

The  $m$  by  $n$  matrix  $A$  is factorized as

$$A = \begin{pmatrix} R & 0 \end{pmatrix} P^H \quad \text{when } m < n,$$

$$A = R P^H \quad \text{when } m = n,$$

where  $P$  is an  $n$  by  $n$  unitary matrix and  $R$  is an  $m$  by  $m$  upper triangular matrix.

$P$  is given as a sequence of Householder transformation matrices

$$P = P_m \cdots P_2 P_1,$$

the  $(m - k + 1)$ th transformation matrix,  $P_k$ , being used to introduce zeros into the  $k$ th row of  $A$ .  $P_k$  has the form

$$P_k = I - \gamma_k u_k u_k^H,$$

where

$$u_k = \begin{pmatrix} w_k \\ \zeta_k \\ 0 \\ z_k \end{pmatrix}.$$

$\gamma_k$  is a scalar for which  $\text{Re } \gamma_k = 1.0$ ,  $\zeta_k$  is a real scalar,  $w_k$  is a  $(k - 1)$  element vector and  $z_k$  is an  $(n - m)$  element vector.  $\gamma_k$  and  $u_k$  are chosen to annihilate the elements in the  $k$ th row of  $A$ .

The scalar  $\gamma_k$  and the vector  $u_k$  are returned in the  $k$ th element of THETA and in the  $k$ th row of A, such that  $\theta_k$ , given by

$$\theta_k = (\zeta_k, \text{Im } \gamma_k).$$

is in THETA( $k$ ), the elements of  $w_k$  are in A( $k, 1$ ), ..., A( $k, k - 1$ ) and the elements of  $z_k$  are in A( $k, m + 1$ ), ..., A( $k, n$ ). The elements of  $R$  are returned in the upper triangular part of A.

### 4 References

Golub G H and van Loan C F (1996) *Matrix Computations* (3rd Edition) Johns Hopkins University Press, Baltimore

Wilkinson J H (1965) *The Algebraic Eigenvalue Problem* Oxford University Press, Oxford

## 5 Parameters

1: M – INTEGER *Input*

*On entry:*  $m$ , the number of rows of  $A$ .

When  $M=0$  an immediate return is effected.

*Constraint:*  $M \geq 0$ .

2: N – INTEGER *Input*

*On entry:*  $n$ , the number of columns of  $A$ .

*Constraint:*  $N \geq M$ .

3: A(LDA,\*) – **complex** array *Input/Output*

**Note:** the second dimension of the array  $A$  must be at least  $\max(1, N)$ .

*On entry:* the leading  $m$  by  $n$  part of the array  $A$  must contain the matrix to be factorized.

*On exit:* the  $m$  by  $m$  upper triangular part of  $A$  will contain the upper triangular matrix  $R$ , and the  $m$  by  $m$  strictly lower triangular part of  $A$  and the  $m$  by  $(n - m)$  rectangular part of  $A$  to the right of the upper triangular part will contain details of the factorization as described in Section 3.

4: LDA – INTEGER *Input*

*On entry:* the first dimension of the array  $A$  as declared in the (sub)program from which F01RJF is called.

*Constraint:*  $LDA \geq \max(1, M)$ .

5: THETA(\*) – **complex** array *Output*

**Note:** the dimension of the array THETA must be at least  $\max(1, M)$ .

*On exit:* THETA( $k$ ) contains the scalar  $\theta_k$  for the  $(m - k + 1)$ th transformation. If  $P_k = I$  then THETA( $k$ ) = 0.0; if

$$T_k = \begin{pmatrix} I & 0 & 0 \\ 0 & \alpha & 0 \\ 0 & 0 & I \end{pmatrix}, \quad \text{Re } \alpha < 0.0$$

then THETA( $k$ ) =  $\alpha$ , otherwise THETA( $k$ ) contains  $\theta_k$  as described in Section 3 and  $\theta_k$  is always in the range  $(1.0, \sqrt{2.0})$ .

6: IFAIL – INTEGER *Input/Output*

*On entry:* IFAIL must be set to 0, -1 or 1. Users who are unfamiliar with this parameter should refer to Chapter P01 for details.

*On exit:* IFAIL = 0 unless the routine detects an error (see Section 6).

For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, for users not familiar with this parameter the recommended value is 0. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**

## 6 Error Indicators and Warnings

If on entry IFAIL = 0 or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL = -1

On entry,  $M < 0$ ,  
or  $N < M$ ,  
or  $LDA < M$ .

## 7 Accuracy

The computed factors  $R$  and  $P$  satisfy the relation

$$(R \ 0)P^H = A + E,$$

where

$$\|E\| \leq c\epsilon\|A\|,$$

$\epsilon$  is the *machine precision* (see X02AJF),  $c$  is a modest function of  $m$  and  $n$ , and  $\|\cdot\|$  denotes the spectral (two) norm.

## 8 Further Comments

The approximate number of floating-point operations is given by  $8m^2(3n - m)/3$ .

The first  $k$  rows of the unitary matrix  $P^H$  can be obtained by calling F01RKF, which overwrites the  $k$  rows of  $P^H$  on the first  $k$  rows of the array A.  $P^H$  is obtained by the call:

```
IFAIL = 0
CALL F01RKF('Separate',M,N,K,A,LDA,THETA,WORK,IFAIL)
```

WORK must be a  $\max(m - 1, k - m, 1)$  element array. If  $K$  is larger than  $M$ , then A must have been declared to have at least  $K$  rows.

Operations involving the matrix  $R$  can readily be performed by the Level 2 BLAS routines F06SJF (CTRSV/ZTRSV) and F06SFF (CTRMV/ZTRMV), (see Chapter F06), but note that no test for near singularity of  $R$  is incorporated into F06SFF (CTRMV/ZTRMV). If  $R$  is singular, or nearly singular then F02XUF can be used to determine the singular value decomposition of  $R$ .

## 9 Example

To obtain the  $RQ$  factorization of the 3 by 5 matrix

$$A = \begin{pmatrix} -0.5i & 0.4 - 0.3i & 0.4 & 0.3 & 0.4i & 0.3i \\ -0.5 & -1.5i & 0.9 - 1.3i & -0.4 - 0.4i & 0.1 - 0.7i & 0.3 - 0.3i \\ -1.0 & -1.0i & 0.2 - 1.4i & 1.8 & 0.0 & -2.4i \end{pmatrix}.$$

### 9.1 Program Text

**Note:** the listing of the example program presented below uses *bold italicised* terms to denote precision-dependent details. Please read the Users' Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

```
*      F01RJF Example Program Text
*      Mark 14 Release.  NAG Copyright 1989.
*      .. Parameters ..
      INTEGER          NIN, NOUT
      PARAMETER        (NIN=5,NOUT=6)
      INTEGER          MMAX, NMAX, LDA
      PARAMETER        (MMAX=10,NMAX=20,LDA=MMAX)
*      .. Local Scalars ..
      INTEGER          I, IFAIL, J, M, N
*      .. Local Arrays ..
      complex         A(LDA,NMAX), THETA(MMAX)
*      .. External Subroutines ..
      EXTERNAL         F01RJF
*      .. Executable Statements ..
      WRITE (NOUT,*) 'F01RJF Example Program Results'
```

```

*      Skip heading in data file
      READ (NIN,*)
      READ (NIN,*) M, N
      WRITE (NOUT,*)
      IF ((M.GT.MMAX) .OR. (N.GT.NMAX)) THEN
        WRITE (NOUT,*) 'M or N is out of range.'
        WRITE (NOUT,99999) 'M = ', M, ' N = ', N
      ELSE
        READ (NIN,*) ((A(I,J),J=1,N),I=1,M)
        IFAIL = 0
*
*      Find the RQ factorization of A
      CALL F01RJF(M,N,A,LDA,THETA,IFAIL)
*
      WRITE (NOUT,*) 'RQ factorization of A'
      WRITE (NOUT,*)
      WRITE (NOUT,*) 'Vector THETA'
      WRITE (NOUT,99998) (THETA(I),I=1,M)
      WRITE (NOUT,*)
      WRITE (NOUT,*)
      + 'Matrix A after factorization (R is in left-hand upper triangle)'
      DO 20 I = 1, M
        WRITE (NOUT,99998) (A(I,J),J=1,N)
20    CONTINUE
      END IF
      STOP
*
99999 FORMAT (1X,A,I5,A,I5)
99998 FORMAT (5(' (',F6.3,',',F6.3,')',:))
      END

```

## 9.2 Program Data

F01RJF Example Program Data

```

3      5      :Values of M and N
( 0.00,-0.50) ( 0.40,-0.30) ( 0.40, 0.00) ( 0.30, 0.40) ( 0.00, 0.30)
(-0.50,-1.50) ( 0.90,-1.30) (-0.40,-0.40) ( 0.10,-0.70) ( 0.30,-0.30)
(-1.00,-1.00) ( 0.20,-1.40) ( 1.80, 0.00) ( 0.00, 0.00) ( 0.00,-2.40)
      :End of matrix A

```

## 9.3 Program Results

F01RJF Example Program Results

RQ factorization of A

Vector THETA

```
( 1.039,-0.101) ( 1.181, 0.381) ( 1.224,-0.000)
```

Matrix A after factorization (R is in left-hand upper triangle)

```
( 0.788, 0.000) (-0.255,-0.401) (-0.277,-0.277) (-0.285, 0.559) ( 0.115, 0.703)
( 0.040, 0.522) (-2.112, 0.000) (-1.109,-0.555) ( 0.128, 0.232) ( 0.079,-0.036)
(-0.227, 0.227) ( 0.045, 0.317) (-3.606, 0.000) ( 0.000,-0.000) ( 0.000, 0.544)
```